Mix Produttivo — Analisi di Sensitività

Un'azienda vuole pianificare il livello di produzione di 3 prodotti (A_1,A_2,A_3) sapendo che la domanda massima è di 4300 pezzi per A_1, 4500 pezzi per A_2 e 5400 pezzi per A_3. Il prezzo di vendita è fissato a 120$ per A_1, 100$ per A_2, 115$ per A_3. Vi sono in totale 22 giorni di produzione disponibili in un mese. Nella seguente tabella sono indicati il costo e la quota di produzione, cioè il massimo numero di pezzi prodotti in 1 giorno se tutte le risorse a disposizione fossero usate per un solo tipo di prodotto.

Produzione	A_1	A_2	A_3
Costo di produzione	$73.30	$52.90	$65.40
Quota di produzione	500	450	550

1. Formulare un modello AMPL per determinare il piano di produzione che massimizza il guadagno totale.
2. A quale prezzo l'azienda sarebbe disposta ad avere un giorno di produzione in più disponibile al mese?
3. Di quanto diminuirebbe il guadagno dell'azienda se la domanda di A_1 diminuisse di 10 pezzi?
4. Di quanto può diminuire la richiesta di pezzi del prodotto A_2 senza che la soluzione cambi?
5. Qual è la somma massima che l'azienda sarebbe disposta ad investire per una campagna pubblicitaria che faccia aumentare la domanda di A_3 di 100 pezzi?

Soluzione

1. Per risolvere il primo punto, formuliamo il problema in termini matematici e poi traduciamo questa formulazione nella sintassi di AMPL.

   (a) FORMULAZIONE.

   - *Indici*: sia i un indice sull'insieme $\{1, 2, 3\}$.
   - *Parametri*:
     - sia P il numero di giorni di produzione in un mese ($P = 22$);
     - sia d_i la domanda massima per il prodotto i ($d_1 = 4300$, $d_2 = 4500$, $d_3 = 5400$);
     - sia v_i il prezzo di vendita per il prodotto i ($v_1 = 120$, $v_2 = 100$, $v_3 = 115$);
     - sia c_i il costo di produzione del prodotto i ($c_1 = 73.30$, $c_2 = 52.90$, $c_3 = 65.40$);
     - sia q_i la quota di produzione massima del prodotto i ($q_1 = 500$, $q_2 = 450$, $q_3 = 550$).
   - *Variabili*: sia x_i la quantità di prodotto i.
     I limiti sulle variabili sono $x_i \geq 0$ e $x_i \in \mathbb{Z}$.
   - *Funzione obiettivo*:
     $$\max \sum_i (v_i - c_i) x_i$$
   - *Vincoli*:
     i. (richiesta): per ogni i, $x_i \leq d_i$ (3 vincoli);
     ii. (produzione): $\sum_i \frac{x_i}{q_i} \leq P$ (1 vincolo).

   (b) MODELLO DI AMPL.

```
# file mixprod.mod - modello per mix produttivo

set PRODOTTI;
param P >= 0;
param domanda { PRODOTTI } >= 0;
param vendita { PRODOTTI } >= 0;
param costo { PRODOTTI } >= 0;
param quota { PRODOTTI } >= 0;

var x { PRODOTTI } >= 0, integer;

maximize guadagno: sum {i in PRODOTTI}
        ( vendita[i] - costo[i]) * x[i];
subject to richiesta {i in PRODOTTI}:
        x[i] <= domanda[i];
subject to produzione:
        sum {i in PRODOTTI} (x[i] / quota[i]) <= P;
```

(c) DATI DI AMPL.

```
# mixprod.dat - file di dati per AMPL - mix produttivo

set PRODOTTI := A1 A2 A3 ;

param P := 22;
param domanda :=
        A1 4300
        A2 4500
        A3 5400 ;
param vendita :=
        A1 120
        A2 100
        A3 115 ;
param costo :=
        A1 73.30
        A2 52.90
        A3 65.40 ;
param quota :=
        A1 500
        A2 450
        A3 550 ;
```

(d) SOLUZIONE DI AMPL. I comandi da dare a AMPL per generare la soluzione di questo problema sono:

```
model mixprod.mod;
data mixprod.dat;
option solver cplex;
solve;
option display_round 4;
display guadagno;
display x;
quit;
```

E la soluzione ottenuta è:

```
CPLEX 7.1.0: optimal integer solution; objective 544531.3
3 MIP simplex iterations
0 branch-and-bound nodes
guadagno = 544531.3000

x [*] :=
A1  4292.0000
A2  1619.0000
A3  5400.0000
;
```

(e) RILASSAMENTO CONTINUO. Poiché non affronteremo il problema dell'analisi di sensitività per problemi con variabili intere, rilassiamo il problema del mix produttivo a variabili di decisione x_i continue per rispondere ai punti (2)-(5) della domanda. Si cambi la riga di dichiarazione delle variabili

```
var x { PRODOTTI } >= 0, integer;
```

con la seguente dichiarazione:

```
var x { PRODOTTI } >= 0;
```

(in pratica, si cancelli la parola-chiave `integer`). La soluzione del rilassamento continuo è:

```
CPLEX 7.1.0: optimal solution; objective 544566.6364
0 simplex iterations (0 in phase I)
guadagno = 544566.6364

x [*] :=
A1  4300.0000
A2  1611.8182
A3  5400.0000
;
```

Si noti che la soluzione del rilassamento continuo è maggiore della soluzione del problema originale (questo succede perché la direzione di ottimizzazione è la massimizzazione); inoltre, non è sufficiente arrotondare la soluzione del rilassamento all'intero più vicino per ottenere la soluzione del problema originale.

(f) VARIABILI SLACK. La forma standard del rilassamento continuo del problema originale prevede l'introduzione di 4 variabili slack (una per ogni vincolo in forma di disequazione). Si immettano i seguenti comandi in AMPL:

```
model mixproduttivo_sensitivita.mod;
data mixproduttivo_sensitivita.dat;
option solver cplex;
solve;
option display_round 4;
```

A questo punto, per ottenere i valori delle variabili di slack relative ai vincoli `richiesta`, si usa il comando:

```
display richiesta.slack;
```

ottenendo i valori:

```
richiesta.slack [*] :=
A1     0.0000
A2  2888.1818
A3     0.0000
;
```

Come si può notare, il valore della variabile di slack del secondo vincolo di richiesta è strettamente positiva: questo significa che non tutta la domanda di prodotto A_2 è soddisfatta. Viceversa, le domande di prodotti A_1 e A_3 sono interamente soddisfatte, dato che le variabili slack dei vincoli corrispondenti valgono 0.

2. La domanda può essere riformulata come "se P aumenta di una unità, come varia il valore della funzione obiettivo?" Per rispondere a questa domanda, dobbiamo ottenere il prezzo ombra del vincolo sui giorni di produzione disponibili, cioè $\sum_i \frac{x_i}{q_i} \leq P$. La variabile duale relativa a questo vincolo si può ottenere da AMPL con il comando

   ```
   display produzione;
   ```

   a cui corrisponde il valore 21195. Questo significa che ogni giorno di produzione disponibile in più aumenta il profitto di altrettanto: dunque l'azienda è disposta a pagare al massimo quel valore per poterne disporre. Questo vale fintantoché la variazione in P non cambia la base ottima.

3. È necessario chiedere ad AMPL di mostrare il valore del prezzo ombra del vincolo di richiesta di A_1:

   ```
   display richiesta;
   ```

   a cui si ottiene un output[1]

   ```
   richiesta [*] :=
   A1   4.3100
   A2   0.0000
   A3  11.0636
   ;
   ```

   Il prezzo ombra del vincolo di richiesta di A_1 è 4.31, per cui ogni aumento unitario della domanda di A_1 aumenta il profitto dell'azienda di altrettanto. Se la domanda diminuisce di 10 pezzi (e se la base ottima non cambia), l'azienda perde 43.1.

4. Il vincolo di richiesta del prodotto A_2 ha prezzo ombra nullo, per cui variazioni nella domanda non alterano il profitto dell'azienda. Questo vale per qualsiasi aumento di domanda e per qualsiasi diminuzione inferiore al valore della corrispondente variabile di slack, cioè inferiore a 2888.1818 pezzi.

5. Il prezzo ombra del vincolo di richiesta su A_3 è 11.0636, per cui ogni aumento unitario della domanda di A_3 fa crescere di altrettanto il profitto dell'azienda. Un aumento di 100 pezzi lo fa crescere di 1106.36. Questa è la massima somma che l'azienda sarebbe disposta a investire per provocare tale aumento.

[1] L'output `richiesta['A3']` a 11.0636 è stato ottenuto con CPLEX 7.1. Con CPLEX 8.1 si ottiene 0 invece di 11.0636. Questo è dovuto a un errore di arrotondamento 'floating point' sulla variabile di slack relativa al vincolo di richiesta di A_3, che viene valutata come 9×10^{-13} anziché 0. Poiché il valore della variabile di slack risulta strettamente positivo (anche se molto vicino a zero), il corrispondente prezzo ombra viene posto a zero. Per evitare questo problema, basta scrivere 5399.999 anziché 5400 nel file di dati.